# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=perl-b-cow
pkgver=0.007
pkgrel=2
pkgdesc="B::COW additional B helpers to check COW status"
arch=('loong64' 'x86_64')
license=('PerlArtistic')
url="https://metacpan.org/release/B-COW"
depends=('perl')
options=('!emptydirs')
source=("https://cpan.metacpan.org/authors/id/A/AT/ATOOMIC/B-COW-$pkgver.tar.gz")
sha512sums=('889e5a57f679735f4f064bc59a7e4c1bec994cb123c83742a165e2ba62a17619e36659b3f0125b6dd6a15616da7100a9d04b66de293a8bf700847d294fe0a2c2')

build() {
  cd B-COW-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd B-COW-$pkgver
  make test
}

package() {
  cd B-COW-$pkgver
  make DESTDIR="$pkgdir" install
}
